

Notes on Hopf Bifurcations, E. Sontag, Sep 03, for Math 613

The Hopf (or “Poincare-Andronov-Hopf”) bifurcation occurs when a pair of complex eigenvalues “crosses the imaginary axis” as a parameter is moved (and, in dimensions, bigger than two, the remaining eigenvalues have negative real part), provided that some additional technical conditions hold. (These conditions tend to be satisfied in examples.) Let us treat the two-dimensional case in detail.

It is best to start with an example:

$$\begin{aligned}\dot{x}_1 &= \mu x_1 - \omega x_2 + \theta x_1(x_1^2 + x_2^2) \\ \dot{x}_2 &= \omega x_1 + \mu x_2 + \theta x_2(x_1^2 + x_2^2)\end{aligned}$$

where $\theta = \pm 1$, or, what is the same, in complex coordinates, using $z = x_1 + ix_2$:


$$\dot{z} = (\mu + \omega i)z + \theta |z|^2 z$$

or, alternatively, in polar coordinates:


$$\begin{aligned}\dot{\rho} &= \rho(\mu + \theta\rho^2) \\ \dot{\phi} &= \omega.\end{aligned}$$

(Note that if we allowed $\theta = 0$, we would have a linear harmonic oscillator, which has no limit cycles. This is basically one of technical conditions that one needs to worry about. On the other hand, allowing any positive [respectively negative] θ would lead to the same behavior as $\theta = 1$ [respectively -1].)

Let us first take the case $\theta = -1$, the “supercritical Hopf bifurcation” case in which we go from a globally asymptotically stable equilibrium to a limit cycle as μ crosses from negative to positive. Since the two polar coordinates can be analyzed separately, and the rotation component is the same no matter what the parameter is, we only need to look at the scalar system $\dot{\rho} = \rho(\mu - \rho^2)$. When $\mu \leq 0$, the origin is the only steady state, and every solution converges to zero (spiraling in). When $\mu \geq 0$, the origin becomes unstable. The right-hand side of the $\dot{\rho}$ equation is negative for $\rho > \sqrt{\mu}$ and positive for $\rho < \sqrt{\mu}$, which means that every trajectory approaches the limit cycle given by the circle of radius $\sqrt{\mu}$ (in complex notation, $z(t) = \sqrt{\mu}e^{i\omega t}$ is the limit cycle). Note that the oscillation has magnitude $\sqrt{\mu}$ and frequency ω .



In the case $\theta = +1$, $\dot{\rho} = \rho(\mu + \rho^2)$. One goes again from stable to unstable as μ goes through zero, but now an *unstable* cycle encircles the origin for $\mu < 0$ (so, the origin is not globally attractive). For $\mu \geq 0$, there is now no cycle that prevents solutions that start near zero from escaping to infinity.



Now suppose given a general system

$$\dot{x} = f(x, \mu)$$

in dimension 2, where μ is a scalar parameter and f is assumed smooth. Suppose that for all μ near zero there is a steady-state $\xi(\mu)$, with eigenvalues $\lambda(\mu) = r(\mu) \pm i\omega(\mu)$, with $r(0) = 0$ and $\omega(0) = \omega_0 > 0$, and that $r'(0) \neq 0$ (“eigenvalues cross the imaginary axis with nonzero velocity”) and that the quantity α defined below is nonzero. Then, up to a local topological equivalence and time-reparametrization, one can reduce the system to the form given in the previous example, and there is a Hopf bifurcation, supercritical or subcritical depending on $\theta =$ the sign of α . (One may interpret the condition on α in terms of a Lyapunov function that guarantees stability at $\mu = 0$, for the supercritical case; see e.g.¹ There is no need to perform the transformation, if all we want is to decide if there is a Hopf bifurcation. The general “recipe” is as follows (see the reference below for proofs).

Let A be the Jacobian of f evaluated at $\xi_0 = \xi(0)$, $\mu = 0$. and find two complex vectors p, q such that

$$Aq = i\omega_0 q, \quad A^T p = -i\omega_0 p, \quad p \cdot q = 1.$$

Compute the dot product $H(z, \bar{z}) = p \cdot F(\xi_0 + zq + \bar{z}\bar{q}, \mu(0))$ and consider the formal Taylor series:

$$H(z, \bar{z}) = i\omega_0 z + \sum_{j+k \geq 2} \frac{1}{j!k!} g_{jk} z^j \bar{z}^k.$$

Then $\alpha = \frac{1}{2\omega_0^2} \text{Re}(ig_{20}g_{11} + \omega_0 g_{21})$.

One may use the following Maple commands, which are copied from “NLDV computer session XI: Using Maple to analyse Andronov-Hopf bifurcation in planar ODEs,” by Yu.A. Kuznetsov,

¹Mees, A.I. Dynamics of Feedback Systems, John Wiley & Sons, New York, 1981.

Mathematical Institute, Utrecht University, November 16, 1999. They are illustrated by the following chemical model (Brusselator):

$$\dot{x}_1 = A - (B + 1)x_1 + x_1^2x_2, \quad \dot{x}_2 = Bx_1 - x_1^2x_2$$

where one fixes $A > 0$ and takes B as a bifurcation parameter. The conclusion is that at $B = 1 + A^2$ the system exhibits a supercritical Hopf bifurcation.

```
restart:
with(linalg):
readlib(mtaylor):
readlib(coeftayl):
F[1]:=A-(B+1)*X[1]+X[1]^2*X[2];
F[2]:=B*X[1]-X[1]^2*X[2];
J:=jacobian([F[1],F[2]],[X[1],X[2]]);
K:=transpose(J);
sol:=solve({F[1]=0,F[2]=0},{X[1],X[2]});
assign(sol);
T:=trace(J);
diff(T,B);
sol:=solve({T=0},{B});
assign(sol);
assume(A>0);
omega:=sqrt(det(J));
ev:=eigenvects(J,'radical');
q:=ev[1][3][1];
et:=eigenvects(K,'radical');
P:=et[2][3][1];
s1:=simplify(evalc(conjugate(P[1])*q[1]+conjugate(P[2])*q[2]));
c:=simplify(evalc(1/conjugate(s1)));
p[1]:=simplify(evalc(c*P[1]));
p[2]:=simplify(evalc(c*P[2]));
simplify(evalc(conjugate(p[1])*q[1]+conjugate(p[2])*q[2]));
F[1]:=A-(B+1)*x[1]+x[1]^2*x[2];
F[2]:=B*x[1]-x[1]^2*x[2];
# use z1 for the conjugate of z:
x[1]:=evalc(X[1]+z*q[1]+z1*conjugate(q[1]));
x[2]:=evalc(X[2]+z*q[2]+z1*conjugate(q[2]));
H:=simplify(evalc(conjugate(p[1])*F[1]+conjugate(p[2])*F[2]));
# get Taylor expansion:
g[2,0]:=simplify(2*evalc(coeftayl(H,[z,z1]=[0,0],[2,0])));
g[1,1]:=simplify(evalc(coeftayl(H,[z,z1]=[0,0],[1,1])));
g[2,1]:=simplify(2*evalc(coeftayl(H,[z,z1]=[0,0],[2,1])));
alpha:=factor(1/(2*omega^2)*Re(I*g[2,0]*g[1,1]+omega*g[2,1]));
evalc(alpha);
# above needed to see that this is a negative number (so supercritical)
```

There are very many references for the topic; I used in particular some material from Yu.A. Kuznetsov. Elements of Applied Bifurcation Theory. 2nd ed., Springer-Verlag, New York, 1998.